Objectif général
Développer une macro VBA permettant de fusionner dynamiquement des onglets issus de plusieurs fichiers Excel, avec une gestion interactive des fichiers, une sélection précise des onglets à fusionner, et des outils de recherche intégrés sur la feuille de fusion.


a. Lancement depuis un fichier vierge (macro)
· L'utilisateur ouvre un fichier Excel vierge contenant uniquement la macro.
· Il peut lancer la macro via un bouton.

b. Sélection des fichiers à traiter
· Une boîte de dialogue s’ouvre :
« Veuillez sélectionner le premier fichier à fusionner »
· L’utilisateur choisit un fichier Excel.
· Une question s’affiche :
« Souhaitez-vous ajouter un autre fichier ? »
· Si oui, l’utilisateur peut sélectionner un nouveau fichier.
· Si non, la procédure passe à l’étape suivante.
· Ce processus se répète jusqu’à ce que l’utilisateur indique ne plus vouloir ajouter de fichiers.

c. Sélection des onglets à fusionner (par fichier)
· Pour chaque fichier sélectionné, une interface (UserForm ou InputBox avancée) affiche la liste des onglets disponibles, tous cochés par défaut.
· L’utilisateur peut décocher ou cocher les onglets qu’il souhaite inclure dans la fusion.
· Les onglets non cochés seront ignorés.

d. Fusion des données
· Les données des onglets sélectionnés sont fusionnées dans un onglet unique nommé « fusion ».
· La fusion respecte les règles suivantes :

· Les en-têtes ne sont copiés qu’une seule fois.
· Une colonne indique l’onglet d’origine, et éventuellement le fichier, si besoin.
· Les données sont ajoutées à la suite, sans écraser les précédentes.
· Les lignes vides sont ignorées.
· Une mise en forme automatique est appliquée (bordures, alternance de couleurs, ajustement des largeurs).

e. Zone de recherche dans l’onglet « fusion »
· Sur la feuille fusion, une zone dédiée est réservée :

· Une cellule (par exemple T2) sert de champ de recherche.
· Un bouton « Rechercher » permet de :

· Filtrer automatiquement les lignes en fonction du critère saisi,
· De conserver uniquement les lignes correspondant à la recherche.
· La recherche peut s’appliquer à une colonne spécifique ou à toutes les colonnes, selon paramétrage.

f. Bouton « Effacer / Réinitialiser »
· Un bouton permet de :

· Vider la zone de recherche,
· Supprimer tous les filtres appliqués,
· Réafficher l’ensemble des lignes fusionnées.

g. Sécurité et ergonomie
· L’ensemble des actions est accompagné de messages clairs (confirmation, fin de traitement).
· La macro gère les erreurs courantes :

· Annulation de la sélection,
· Fichiers déjà ouverts,
· Onglets manquants,
· Absence de données.




La macro sur laquelle j’avais travaillée et que je souhaiterais améliorer : 

Sub fusion_onglets()

    Dim sh As Worksheet
    Dim dest As Worksheet
    Dim LastRow As Long, ligneStop As Long
    Dim plage As Range, rng As Range
    Dim nextRow As Long, lastDestRow As Long
    Dim headerDone As Boolean
    Dim r As Long
    Dim col As Range

    Application.ScreenUpdating = False
    Application.EnableEvents = False

    Set dest = Sheets("fusion")

    '----------------------------------------------------------
    ' 0) RÉINITIALISATION DES LIGNES APRÈS LA 4e
    '----------------------------------------------------------
    dest.Rows("4:" & dest.Rows.Count).ClearContents
    dest.Rows("4:" & dest.Rows.Count).Interior.Pattern = xlNone
    dest.Rows("4:" & dest.Rows.Count).Borders.LineStyle = xlNone

    headerDone = False

    '----------------------------------------------------------
    ' 1) DATE DE FUSION
    '----------------------------------------------------------
    dest.Range("A2").Value = "Fusion réalisée le " & Format(Now, "dd/mm/yyyy à HH:MM:ss")
    dest.Range("A2").Font.Bold = True
    dest.Range("A2").Font.Color = vbBlue

    '----------------------------------------------------------
    ' 2) SUPPRESSION DES LIGNES VIDES DES ONGLET SOURCES
    '----------------------------------------------------------
    For Each sh In Worksheets
        If sh.Name <> dest.Name Then

            LastRow = sh.Cells(sh.Rows.Count, "A").End(xlUp).Row

            If LastRow >= 9 Then
                Set rng = sh.Range("A9:R" & LastRow)

                On Error Resume Next
                rng.SpecialCells(xlCellTypeBlanks).EntireRow.Delete
                On Error GoTo 0

            End If
        End If
    Next sh

    '----------------------------------------------------------
    ' 3) COPIE DES ENTÊTES (UNE SEULE FOIS)
    '----------------------------------------------------------
    For Each sh In Worksheets
        If sh.Name <> dest.Name And Not headerDone Then

            sh.Range("A1:R8").Copy
            dest.Range("B1").PasteSpecial xlPasteValues
            dest.Range("B1").PasteSpecial xlPasteFormats
            Application.CutCopyMode = False

            ' Ajout colonne A (onglet origine)
            dest.Range("A8").Value = "onglet_origine"
            dest.Range("A8").Font.Bold = True
            dest.Range("A8").Interior.Color = RGB(204, 232, 255) ' bleu ciel
            dest.Range("A8").HorizontalAlignment = xlCenter

            headerDone = True
        End If
    Next sh

    '----------------------------------------------------------
    ' 4) COPIE DES DONNÉES + NOM ONGLET
    '----------------------------------------------------------
    For Each sh In Worksheets
        If sh.Name <> dest.Name Then

            LastRow = sh.Cells(sh.Rows.Count, "A").End(xlUp).Row

            If LastRow >= 9 Then

                ligneStop = IIf(LastRow > 5000, 5000, LastRow)
                Set plage = sh.Range("A9:R" & ligneStop)

                nextRow = dest.Cells(dest.Rows.Count, "B").End(xlUp).Row + 1
                If nextRow < 9 Then nextRow = 9

                dest.Cells(nextRow, 2).Resize(plage.Rows.Count, plage.Columns.Count).Value = plage.Value
                dest.Cells(nextRow, 1).Resize(plage.Rows.Count, 1).Value = sh.Name

            End If
        End If
    Next sh

    '----------------------------------------------------------
    ' 5) MISE EN FORME GÉNÉRALE
    '----------------------------------------------------------
    lastDestRow = dest.Cells(dest.Rows.Count, "A").End(xlUp).Row

    If lastDestRow >= 8 Then

        dest.Range("A8:S" & lastDestRow).VerticalAlignment = xlTop
        dest.Range("A1:S" & lastDestRow).Font.Size = 9
        dest.Range("A8:S" & lastDestRow).WrapText = True

        With dest.Range("A8:S" & lastDestRow).Borders
            .LineStyle = xlContinuous
            .Weight = xlThin
        End With

        ' Couleurs alternées
        For r = 9 To lastDestRow Step 2
            dest.Range("A" & r & ":S" & r).Interior.Color = RGB(235, 235, 235)
        Next r

        dest.Columns("A:S").EntireColumn.AutoFit

        dest.Columns("K").ColumnWidth = 30

        For Each col In dest.Columns("A:S")
            If col.ColumnWidth > 60 Then col.ColumnWidth = 60
            If col.ColumnWidth < 8 Then col.ColumnWidth = 8
        Next col

        dest.Rows("8:" & lastDestRow).EntireRow.AutoFit

        For r = 8 To lastDestRow
            If dest.Rows(r).RowHeight < 12 Then dest.Rows(r).RowHeight = 12
        Next r

    End If

    '----------------------------------------------------------
    ' 6) FIGER : Ligne 8 + colonne A
    '----------------------------------------------------------
    dest.Range("B9").Select
    ActiveWindow.FreezePanes = False
    ActiveWindow.FreezePanes = True

    '----------------------------------------------------------
    ' 7) FILTRE
    '----------------------------------------------------------
    dest.Range("A8").CurrentRegion.AutoFilter

    Application.EnableEvents = True
    Application.ScreenUpdating = True

    MsgBox "Fusion terminée !", vbInformation

End Sub



